Computer Structure and Language

Hamid Sarbazi-Azad
Department of Computer Engineering
Sharif University of Technology (SUT)
Tehran, Iran

IBM System 360/370:

- IBM360 was introduced in 1964 (IBM370 in 1970); a CISC machine that could perform about 35 KIPS (compatible to USSR's Ryad).
- IBM370 has 13 instructions more than IBM360.
- In 1989, IBM360/370 machines accounted for more than half of the estimated $260B value of large computer systems worldwide.

Machine Structure:

Memory size:	2^{24} bytes
Addressable unit:	8 bits (byte)
Word length:	32 bits, Big Endian, Aligned.
ISA types:	Register-Register, Register-Memory, Memory-Memory.
Addressing modes:	Segmented, Indexed, Register (direct/indirect), Implied, and Immediate.
Data types:	Signed integer (2's complement), Unsigned binary, Character (byte), Decimal, Floating-point.

Internal Registers:



Base-Displacement Addressing:

Storage address = (**B**ase register) + **D**isplacement



Data Alignment in Main Memory:



Data Alignment in Main Memory (cont.):



Instruction Formats:

RR (**R**egister-**R**egister): 2 Bytes



RX (**R**egister-Inde**x**ed): 4 Bytes

RS (**R**egister-**S**torage): 4 Bytes

SI (**S**torage-**I**mmediate): 4 Bytes

SS1 (**S**torage-**S**torage **1**-Length): 6 Bytes

SS2 (**S**torage-**S**torage **2**-Length): 6 Bytes

Instruction Formats:

RR (**R**egister-**R**egister)



All mnemonics of this format have character R at the end.

Example 1:
Assembly instruction: SR 12,5
Operation: R12 ← (R12) - (R5)
Machine code: 1BC5

Example 2:
Assembly instruction: NR 6,10
Operation: R6 ← (R6) ∧ (R10)
Machine code: 146A

Instruction Formats:

RX (**R**egister-Inde**x**ed)

Most machine instructions are coded in this format.

Example 1:
Assembly instruction: S 10,NUM
Operation: R10 ← (R10) - (M_{NUM})
Machine code: 5BA0C01A

NUM address = (R12) + 01Ah

Example 2:
Assembly instruction: L 15,N10
Operation: R15 ← (M_{N10});
Machine code: 58F0B11B

N10 address = (R11) + 11Bh

Instruction Formats:

RS (**R**egister-**S**torage)

This format includes some 3-operand instructions.

Example 1:
Assembly instruction: SLL 5,3
Operation: R5 ← (R5) <<3
Machine code: 89500003

Example 2:
Assembly instruction: LM 2,5,A
Operation: R2←(M_A); R3←(M_{A+4}); R4←(M_{A+8}); R5←(M_{A+12});
Machine code: 9825C01B

A address = (R12) + 01Bh

Instruction Formats:

SI (**S**torage-**I**mmediate)

Only byte immediate can be used in this format. We will see later how programmer can use other immediate data types.

Example 1:
Assembly instruction: MVI T1,X'5F'
Operation: M_{T1} ← 5F
Machine code: 925FC0BA

T1 address = (R12) + 0BAh

Example 2:
Assembly instruction: OI 100(3),C'*'
Operation: $M_{(R3)+100}$←($M_{(R3)+100}$) v 5C;
Machine code: 965C3064

EBCDIC('*') = 5Ch

Instruction Formats:

SS1 (**S**torage-**S**torage **1**-Length)

Example 1:

Assembly instruction: MVC DST(100),SRC

Operation: Move 100 bytes from M_{SRC} to M_{DST}.

Machine code: D263C00AC110

DST address = (R12) + 0Ah
SRC address = (R12) +110h

Example 2:

Assembly instruction: OC AREA(10),10(4)

Operation: Or 10 bytes in M_{AREA} with 10 byte from $M_{10+(R4)}$

Machine code: D609911B400A

AREA address = (R9)+11Bh

Instruction Formats:

SS2 (**S**torage-**S**torage **2**-Length)

Example:

Assembly instruction: AP NUM1(6),NUM2(5)

Operation: Add decimal number of 5 bytes at M_{NUM2} to a decimal number of 6 bytes in $M_{NUM1.}$

Machine code: FA54C10ACF10

NUM1 address = (R12) + 10Ah
NUM2 address = (R12) + F10h

Program Structure:

```
PROG-NAME      START    Initial Address        → Default Initial Address is 0
                .
                .
START-LABEL    ...
                .
                .
   Some directives/instructions to define Base register and initialize it
                .
                .
   Instructions
                .
                .
   Some instructions to return the control to OS
                .
                .
   Some directives to define variables
                .
                .
                END      START-LABEL
```


Notations:

- General format of an assembly instruction is:

Label **Mnemonic** **Opr1,Opr2,Opr3** **Inline Comment can come after last operand**

Note that:

- Labels should start at first column.
- Full-line comment starts with * at the first column
- Mnemonic come at column 2+.
- Label, Mnemonic, Operands and then Inline comment are separated by 1+ blanks.

Example:

```
LOOP    AR      5,6    Add the content of R6 into R5
* "LOOP" is a label, "AR" is mnemonic, "5,6" show operands, and  "Add the content of R6 to R5"
* is an inline comment
  L 6,ADAD1
*  No label is used, "L" is mnemonic, "6,ADA1" show operands, no comment is used
```

Notations:

You can use:

- Numbers in the assembly program in (default) decimal, Octal, Binary and Hexadecimal bases using O, B, and X prefixes.

Example: B'000001010111" == O'0127" == X'057' == 87

- **DS** (define storage) directive to define a variable (allocate space for it)

Example: ARRAY DS 20F

- **DC** (define constant) directive to define a variable and initialize it

Example: NUM1 DC F'-5'

- **EQU** (Equate) directive to define a constant.

Example: R5 EQU 5 then use as: AR R5,R5 == AR 5,5

Examples of Variable Definition:

```
.....
* We define some WORD variables.
* Lets assume Location Counter = 0000FEh, here.
VAR1    DS      3F
VAR2    DC      F'1'
VAR3    DC      F'1',F'-2',2F'100'
.....
```

Main Memory



Symbol Table

Symbol	Address
VAR1	000100h
VAR2	00010Ch
VAR3	000110h

Sample Program:

```
STOCK    START 256
BEGIN    BALR  11,0
         USING *,11
         L     3,OLDOH
         A     3,RECPT
         S     3,ISSUE
         ST    3,NEWOH
         EOJ
OLDOH    DC    F'9'
RECPT    DC    F'4'
ISSUE    DC    F'6'
NEWOH    DS    F
```

```
  LOC   OBJECT CODE     ADDR1 ADDR2  STMT   SOURCE STATEMENT

000100                               1 STOCK    START 256
000100 05B0                          2 BEGIN    BALR  11,0
000102                               3          USING *,11
000102 5830 B012              00114  4          L     3,OLDOH
000106 5A30 B016              00118  5          A     3,RECPT
00010A 5B30 B01A              0011C  6          S     3,ISSUE
00010E 5030 B01E              00120  7          ST    3,NEWOH
                                     8          EOJ
000114 00000009                      9 OLDOH    DC    F'9'
000118 00000004                     12 RECPT    DC    F'4'
00011C 00000006                     13 ISSUE    DC    F'6'
000120                              14 NEWOH    DS    F
000100                              15          END   BEGIN
```

The listing of the program

End of Slides